
Características generales

 Modelo básico
– planificación con prioridades dinámicas

» también EDF
– políticas de despacho para tareas con la misma prioridad

» FIFO, RR
– políticas de acceso a objetos protegidos

» ICPP / SRP

 Perfil de Ravenscar
– FPPS + ICPP
– modelo de tareas estático, ejecución previsible
– se puede analizar el comportamiento temporal

 En el anexo de tiempo real se define un modelo de
planificación con prioridades y desalojo

 La prioridad de una tarea es de un subtipo definido en el
paquete System

Prioridades

subtype Any_Priority is Integer range intervalo;
subtype Priority is Any_Priority
  range Any_Priority’First .. valor;
subtype Interrupt_Priority is Any_Priority
  range Priority’Last + 1 .. Any_Priority’Last;

 Los valores mayores denotan prioridades más altas
 Debe haber al menos 30 valores de Priority

y 1 de Interrupt_Priority

Prioridad básica

 La prioridad básica de una tarea es la prioridad con la que
se ejecuta inicialmente, si no se dice otra cosa

 Se especifica con un pragma Priority

task Controller is
   pragma Priority (10);
end Controller;

task type Sensor (Task_Priority : System.Priority) is
  pragma Priority (Task_Priority);
end Sensor;

 La prioridad básica de una tarea se puede cambiar con
Ada.Dynamic_Priorities.Set_Priority

Prioridad activa

 Una tarea puede heredar una prioridad más alta que su
prioridad básica
– durante la activación hereda la prioridad del padre
– durante la ejecución de una operación protegida hereda el techo de

prioridad del objeto protegido
– durante la ejecución de una cita la tarea llamada hereda la prioridad

de la que llama

 La prioridad activa  de una tarea es el máximo de su
prioridad básica y cualquier prioridad heredada
– esta es la prioridad que se usa para planificar el uso del procesador

Modelo de planificación

 Hay una cola de tareas listas
(ready queue) para cada nivel de
prioridad

– es un modelo, no tiene por qué
realizarse exactamente así

 Cuando el procesador está libre se
despacha la primera tarea de la
cola de mayor prioridad que no
esté vacía

 Cuando se llega a un punto de
despacho la tarea que está en
ejecución se vuelve a poner en la
cola correspondiente a su prioridad
activa

– por ejemplo, si la tarea se
suspende

…
…

suspended

running

suspend

other
dispatching

events

ready

resume

Políticas de planificación

 Definen cómo se hace la planificación en detalle
– cuáles son los puntos de despacho
– cómo se gestionan las colas

 El pragma

define la política de planificación para todo el sistema
– debe ser una de las siguientes

» FIFO_Within_Priorities
» Non_Preemptive_FIFO_Within_Priorities
» Round_Robin_Within_Priorities
» EDF_Across_Priorities

pragma Task_Dispatching_Policy (policy)

Planificación con desalojo

 FIFO_within_priorities es una política de planificación con
prioridades fijas y desalojo (FPPS)

 Puntos de despacho
– reanudación de una tarea suspendida

» la tarea reanudada se pone al final de la cola correspondiente a su
prioridad activa

– ejecución de un delay que no da lugar a un retardo
» la tarea en ejecución se pone al final de la cola correspondiente a su

prioridad activa
– activación de una tarea con mayor prioridad

» la tarea en ejecución es desalojada y se pone al principio de la cola
correspondiente a su prioridad activa

Planificación sin desalojo

 Non_Preemptive_FIFO_within_priorities es una política de
planificación con prioridades fijas y sin desalojo (FPNS)

 Puntos de despacho
– reanudación de una tarea suspendida

» la tarea reanudada se pone al final de la cola correspondiente a su
prioridad activa

– ejecución de un delay que no da lugar a un retardo
» la tarea en ejecución se pone al final de la cola correspondiente a su

prioridad activa

Turno circular

 Round_Robin_Within_Priorities es una política de
planificación en la que se efectúa un turno circular (RR)
entre las tareas de la misma prioridad

 Como FIFO_within_priorities, excepto:
– cuando una tarea se pone al final de una cola se le asigna un

presupuesto de tiempo de ejecución igual a un cierto cuanto
» por ejemplo, cuando se reanuda después de estar suspendida

– cuando la tarea se ejecuta, su presupuesto disminuye según el
tiempo de ejecución que consume

– cuando una tarea se pone al principio de una cola mantiene el
presupuesto que le queda

» por ejemplo, cuando se la desaloja
– cuando una tarea ha consumido todo su presupuesto se pone al

final de la cola correspondiente a su nivel de prioridad (y se
despacha otra tarea)

» excepto si está en una operación protegida

Ada.Dispatching.Round_Robin

with System;
with Ada.Real_Time;

package Ada.Dispatching.Round_Robin is

  Default_Quantum : constant Ada.Real_Time.Time_Span
    := implementation-defined;

  procedure Set_Quantum (Pri       : in System.Priority;
                         Quantum   : in Ada.Real_Time.Time_Span);
  procedure Set_Quantum (Low, High : in System.Priority;
                         Quantum   : in Ada.Real_Time.Time_Span);
  function Actual_Quantum (Pri : System.Priority)

 return Ada.Real_Time.Time_Span;
  function Is_Round_Robin (Pri : System.Priority) return Boolean;

end Ada.Dispatching.Round_Robin;

Primero el más urgente

 EDF_Across_Priorities es una política de planificación
basada en plazos

 El plazo de respuesta de una tarea se especifica con

El plazo es una expresión de tipo Ada.Real_Time.Time_Span

 Cada tarea tiene una prioridad y un plazo relativo
– cuando está activa tiene, además, un plazo absoluto

 Se despacha primero la tarea cuyo plazo absoluto es
menor, independientemente de su prioridad

pragma Relative_Deadline (plazo)

Ada.Dispatching.EDF

with Ada.Real_Time;
with Ada.Task_Identification;

package Ada.Dispatching.EDF is

  subtype Deadline is Ada.Real_Time.Time;
  Default_Deadline : constant Deadline := Ada.Real_Time.Time_Last;

  procedure Set_Deadline (
     D : in Deadline;

  T : in Ada.Task_Identification.Task_Id
            := Ada.Task_Identification.Current_Task);

  procedure Delay_Until_And_Set_Deadline (
  Delay_Until_Time : in Ada.Real_Time.Time;
  Deadline_Offset  : in Ada.Real_Time.Time_Span);

  function Get_Deadline (
  T : in Ada.Task_Identification.Task_Id :=
              Ada.Task_Identification.Current_Task)
return Deadline;

end Ada.Dispatching.EDF;

Ejemplo (1)

-- gnat.adc:
-- pragma Task_Dispatching_Policy (EDF_Across_Priorities);

with Ada.Real_Time;

package Deadline_Driven is
  use Ada.Real_Time;

  Deadline : constant Time_Span := Milliseconds (4);

  task Periodic is
    pragma Relative_Deadline (Deadline);
  end Periodic;

end Deadline_Driven;

Ejemplo (2)

with Ada.Real_Time;
with Ada.Dispatching.EDF;

package body Deadline_Driven is
  use Ada.Real_Time, Ada.Dispatching.EDF;

  task body Periodic is
    Period : constant Time_Span := Milliseconds (20);
    Next   : Time := Clock;
  begin
    loop
      Delay_Until_And_Set_Deadline (Next, Deadline);
      -- actividad periódica
      Next := Next + Period;
    end loop;
  end Periodic;

end Deadline_Driven;

Techo de prioridad

 La prioridad activa de las tareas que ejecutan operaciones
protegidas depende de la política de planificación y de la
política de acceso a los objetos protegidos

 La política de acceso con techos de prioridad (ceiling
locking) está definida para todas las políticas de
planificación predefinidas
– con FIFO_Within_Priorities equivale al protocolo del techo de

prioridad inmediato
– con EDF_Across_Priorities equivale al protocolo SRP (stack

resource protocol)

Prioridad techo

 Para usar techos de prioridad hay que poner un pragma
de configuración

– afecta a todo el programa (con GNAT se pone en gnat.adc)

 El techo de prioridad de un objeto o un tipo protegido se
especifica mediante un pragma Priority

– el valor por omisión es Priority’Last

pragma Locking_Policy (Ceiling_Locking);

protected type Resource (Ceiling : System.Priority)is
   ...
   pragma Priority (Ceiling);
end Resource;

Ejecución de operaciones protegidas

 Cuando se usa el protocolo del techo de prioridad
inmediato se pueden implementar las barreras de forma
muy eficiente (proxy model):

– cuando termina un procedimiento o entrada protegida se ejecutan
las entradas que se puedan haber abierto como consecuencia

– esto se hace con el mismo flujo de control de la tarea que llama a
la primera operación

– se reduce el número de cambios de contexto
– todo esto es posible porque las operaciones se ejecutan con

prioridad elevada

Ejemplo

protected Gate_Control is
  pragma Priority(28);
  entry Stop_And_Close;
  procedure Open;
private
  Gate : Boolean := False;
end Gate_Control;

 Supongamos que
– una tarea T (prioridad = 20)

llama a Stop_and_Close y se
suspende

– otra tarea S (prioridad =27)
llama a Open

 La hebra de S ejecuta:
– el código de Open
– la evaluación de la barrera
– el código de Stop_and Close

(por cuenta de T)
– la evaluación de la barrera otra

vez
– la vuelta al punto donde S llamó

a Open
 No hay cambio de contexto

protected body Gate_Control is
  entry Stop_And_Close
    when Gate is
  begin
    Gate := False;
  end;
  procedure Open is
  begin
    Gate := True;
  end;
end Gate_Control;

Prioridades dinámicas

 La prioridad base de una tarea se puede cambiar
dinámicamente

 El techo de prioridad de un objeto protegido también se
puede cambiar dinámicamente

Ada.Dynamic_Priorities.Set_Priority (…);

Resource’Priority := …

Ada.Dynamic_Priorities

with System;
with Ada.Task_Identification;
package Ada.Dynamic_Priorities is
  procedure Set_Priority
    (Priority : in System.Any_Priority;
     T        : in Ada.Task_Identification.Task_Id :=
                      Ada.Task_Identification.Current_Task);

  function Get_Priority
    (T : Ada.Task_Identification.Task_Id :=

         Ada.Task_Identification.Current_Task)
 return System.Any_Priority;

end Ada.Dynamic_Priorities;

Política de gestión de colas

 Las colas de las entradas se pueden ordenar
– por orden de llegada

– por orden de prioridades

 El pragma Queuing_Policy es un pragma de configuración

pragma Queuing_Policy (FIFO_Queuing);

pragma Queuing_Policy (Priority_Queuing);

Perfil de Ravenscar

 El perfil de Ravenscar impone las políticas
FIFO_Within_Priorities y Ceiling_Locking
– FPPS con ICPP

 Algunas cosas están prohibidas
– prioridades y techos de prioridad dinámicos

 El motivo es asegurar que se puede analizar el
comportamiento temporal de los sistemas

6
Resumen

 Ada define un modelo de planificación flexible y completo
– modelo básico: prioridades
– detalles: FPPS, FPNS, RRS, EDF
– objetos protegidos: ICPP, SRP

 El perfil de Ravenscar restringe este modelo para poder
analizar el comportamiento temporal


